

Report: Decentralized Inheritance Protocol

Noah Klaholz, Vincent Schall, Max Mendes Carvalho

November 2025

Contents

1	Introduction	2
1.1	Motivation	2
1.2	The Decentralized Inheritance Protocol	2
2	Smart Contract architecture	2
2.1	General Design and Flow	2
2.2	Roles and Access Control	3
2.3	Beneficiaries and Payout Logic	3
2.4	Funds Management and Aave Integration	3
2.5	Death Verification and Oracles	4
2.6	Security Considerations	4
2.7	Gas and Scalability	4
2.8	Known Limitations and Future Work	4
3	Design Decisions and Scope Refinement	5
3.1	Removal of Vesting Schedules	5
3.2	Manual State Transitions Over Automated Keepers	6
3.3	Aave Integration for Yield Generation	6
3.4	Single Notary vs. Multi-Signature Verification	7
3.5	Scope Reductions and Future Work	7
3.6	Summary of Trade-offs	7
4	Tool usage / tech stack	7
4.1	Core Languages	7
4.2	Smart Contract Framework and Tooling	7
4.3	Testing Stack	8
4.4	Linting and Code Quality	8
4.5	Runtime and Scripting	8
	Appendices	8
A	References	17

1 Introduction

1.1 Motivation

No one can escape death - but what happens to your crypto when you die? According to [3], it is estimated that around 3.7 million Bitcoin are lost and unrecoverable. One of the top reasons is death: crypto holders that passed away and failed to share access information with heirs will be responsible for inaccessible funds.

Traditional inheritance systems are flawed: they take very long, are expensive and more often than not lead to conflict between the heirs. We want to solve these problems by introducing a decentralized inheritance protocol.

1.2 The Decentralized Inheritance Protocol

The idea is as follows: anyone can create a will by deploying the inheritance protocol contract. After that, depositing coins, tokens and assets, as well as defining beneficiaries or heirs by adding their wallet addresses, is quick and easy with function calls to the contract. For each beneficiary, the owner can define a payout amount as a percentage of the total deposited assets.

Furthermore, deposited assets are invested using Aave¹. This allows the balance to grow instead of laying dry.

The owner has to check in at least every 90 days to verify that he's still alive. As long as these check-ins occur, there will be no payout. When a check-in is missed, there is a 30-day grace period during which a check-in can still be made to reset the timer. If no check-in occurs within this grace period, the contract enters a verification phase where a trusted notary can upload death verification.

In case of death, trusted oracles (in most cases a notary) are used to verify the death via death certificates before initiating the payout.

2 Smart Contract architecture

2.1 General Design and Flow

The inheritance protocol is implemented in a single smart contract and composes well-known primitives from *OpenZeppelin* for access control and safety [15]. In particular, the contract inherits from *Ownable* to grant the will's creator administrative privileges, uses *ReentrancyGuard* to protect sensitive functions, and interacts with funds through the *IERC20* interface of the ERC-20 standard [4]. For external integration, the contract talks to a lending pool (Aave-compatible mock) to invest idle balances and to a death oracle for verification.

The constructor wires all dependencies (token, death oracle, notary, pool) and initializes the state machine that models the life cycle of a will. The state machine transitions among four phases:

```
1 enum State { ACTIVE, WARNING, VERIFICATION, DISTRIBUTION }
```

Listing 1: Contract state machine

ACTIVE is the normal operating phase where the owner can manage beneficiaries and funds. If the owner misses a check-in for more than 90 days, the state moves to *WARNING*. After a 30-day grace period without check-in, the state advances to *VERIFICATION*. Once the death oracle confirms the owner's passing, the state becomes *DISTRIBUTION*, which triggers payout.

¹Aave — a decentralized lending protocol: supply crypto to earn interest via liquidity pools. <https://aave.com/docs/developers/liquidity-pool>

The `updateState()` function is public so that the notary, family members, or any third party can progress the state machine when the objective conditions are met. This creates an incentive-aligned mechanism: beneficiaries want the state to be up to date to receive their funds and a trusted notary can be instructed to call this function regularly. When the state reaches *DISTRIBUTION*, the contract immediately invokes `distributePayout()` and emits a `StateChanged` event.

2.2 Roles and Access Control

We use `Ownable` for a single privileged owner (the testator), and a dedicated *notary* address for external verification tasks [15]. Access is enforced by modifiers:

- `onlyOwner`: administrative actions (check-in, adding/removing beneficiaries, deposits/withdrawals) are restricted to the owner.
- `onlyNotary`: only the notary can upload death verification proofs.
- `onlyPreDistribution`: prevents fund mutations once the system is in the distribution phase.
- `onlyDistribution`: guards payout functions so they are callable only in the final phase.
- `onlyActiveWarning`: Prevents Administrative changes like adding beneficiaries from being executed unless in the `ACTIVE` or `WARNING` phase.

Functions that transfer value also use the `nonReentrant` modifier from `ReentrancyGuard` to mitigate reentrancy (SWC-107) [1].

2.3 Beneficiaries and Payout Logic

Beneficiaries are kept in a fixed-size array of at most ten entries to keep gas costs predictable and iteration bounded. Each entry stores a payout address and a percentage amount. The contract enforces that:

- No duplicate beneficiary addresses exist.
- The total determined payout never exceeds 100%.
- Add/remove operations are only allowed before distribution and require a fresh owner check-in.
- All administrative changes can only be made by the contract's owner.

On distribution, the contract retrieves the pool balance from Aave, computes each beneficiary's share by percentage, and transfers tokens accordingly by iterating through the list of beneficiaries. If the sum of percentages is below 100%, the residual is sent to a donation address to prevent funds from being stranded forever.

2.4 Funds Management and Aave Integration

The protocol accepts an ERC-20 token (MockUSDC in our deployment) via `deposit`. The owner first approves the contract to spend tokens, then the contract supplies tokens into an Aave-compatible pool (MockAavePool in our deployment) to accrue yield [4, 2]. Withdrawals reverse the flow: tokens are pulled from the pool and transferred back to the owner. Critical operations are protected with `nonReentrant` and disallowed after distribution.

2.5 Death Verification and Oracles

Death verification is abstracted behind the `IDeathOracle` interface. The notary calls `uploadDeathVerification` with a boolean and proof bytes; the oracle persists the attestation and can still be called upon by beneficiaries to verify the death. The state machine polls the oracle by calling `isDeceased(owner())` and, if true, transitions to *DISTRIBUTION*. In our test setup we use a mock oracle to enable deterministic unit tests. For a production deployment, this component could be backed by a notarized registry, a government API gateway, or decentralized oracle networks. Additionally, `updateState()` could be automated using off-chain keepers (e.g., Chainlink Automation) to guarantee timely transitions without relying on manual calls [6]. However, since beneficiaries already have an incentive to update the state regularly, the decision was made to avoid the extra cost for off-chain automation.

2.6 Security Considerations

Our design follows standard Solidity best practices [16, 15]:

- Reentrancy protection on functions that transfer tokens [1].
- Access control via explicit roles and clear phase guards (`onlyPreDistribution`, `onlyDistribution`, ...).
- Use of `immutable` and `constant` for critical configuration to reduce runtime risk and gas cost.
- Bounded iteration over at most ten beneficiaries to avoid unbounded gas usage.
- Overflow/underflow safety from the Solidity 0.8.x checked arithmetic [16].

Threats and mitigations:

- **Oracle risk:** a compromised notary/oracle could wrongfully trigger distribution. This is mitigated organizationally (trusted notaries) and could be strengthened with multi-sig attestations or time delays.
- **Griefing/liveness:** anyone can call `updateState()`, but transitions are conditional and idempotent; no value is at risk.
- **External calls:** interactions with the pool and token are performed after state updates and protected by `nonReentrant`. The donation transfer happens last to simplify reasoning.

2.7 Gas and Scalability

The fixed-size array avoids storage resizes and bounds loops to a maximum of ten iterations. Getter functions such as `getActiveBeneficiaries()` build a compact memory array for off-chain consumers, trading a small amount of gas for simpler client logic. State checks in `updateState()` are in constant time. While the design targets personal wills (low on-chain scale), it remains economical for typical usage.

2.8 Known Limitations and Future Work

- **Single-asset support:** the current implementation handles one ERC-20 token instance. Extending to multiple assets would require per-asset accounting and distribution.
- **Maximum of ten beneficiaries:** chosen for simplicity and predictable gas; a dynamic structure[7] could be introduced. However this would require careful consideration of gas efficiency and time complexity, while larger will structures could also be realized by having multiple wills and distributing funds accordingly.

- **Oracle centralization:** production setups should consider decentralized attestations or multi-party notaries.
- **Automation:** integrating keepers would remove the need for manual `updateState()` calls [6].
- **UX improvements:** support for EIP-2612 `permit` and richer events to make indexing easier [8].

3 Design Decisions and Scope Refinement

The initial project proposal outlined an ambitious feature set including vesting schedules, aggregated releases, multi-asset support, and automated state transitions. During development, we refined the scope based on technical feasibility, cost-benefit analysis, and project timeline constraints. This section documents the key decisions and trade-offs.

3.1 Removal of Vesting Schedules

The original design included sophisticated vesting mechanisms to distribute inheritances gradually over time (e.g., 10% per year for 10 years) to prevent sudden wealth syndrome and protect young beneficiaries from mismanaging lump sums. However, we ultimately removed vesting from the final implementation.

Technical Challenge: Time-Based Execution. Smart contracts on Ethereum are fundamentally passive: they cannot self-execute based on time. A vesting schedule requires periodic unlocking (e.g., monthly or yearly), but no on-chain primitive can trigger a function call at a future timestamp without external intervention. This creates three potential solutions, each with significant drawbacks:

1. **Off-chain automation (Chainlink Keepers):** Use a keeper network to call `releaseVestedFunds()` at scheduled intervals [6]. This introduces recurring costs (keeper fees), centralizes liveness assumptions (keeper availability), and adds complexity to the deployment and maintenance workflow.
2. **Pull-based vesting:** Beneficiaries call `withdraw()` whenever they want funds, and the contract calculates how much has vested since the last withdrawal. While this eliminates automation costs, it creates poor UX (beneficiaries must remember to claim) and removes the protective aspect of enforced gradual distribution—beneficiaries can simply wait and withdraw the full amount later.
3. **Incentivized third-party calls:** Reward external actors for calling unlock functions by giving them a small fee. This adds game-theoretic complexity, potential griefing vectors (claiming fees without benefiting beneficiaries), and still requires someone to monitor and trigger transactions.

Scope Prioritization. Given a fixed development timeline, we chose to focus on the core inheritance mechanism (state machine, death verification, percentage-based distribution) rather than vesting. The current design delivers immediate value for the primary use case—preventing total loss of crypto assets upon death—while vesting primarily addresses a secondary concern (beneficiary financial responsibility). Future iterations could introduce optional vesting as a modular extension for users who specifically need it.

Alternative Mitigations. Users concerned about lump-sum distributions can approximate vesting by creating multiple inheritance contracts with staggered check-in periods or by allocating funds to custodial services that provide off-chain vesting. Additionally, the protocol’s percentage-based allocation allows splitting inheritances among trustees or financial advisors who can manage gradual distributions off-chain.

3.2 Manual State Transitions Over Automated Keepers

The contract’s state machine (`ACTIVE` → `WARNING` → `VERIFICATION` → `DISTRIBUTION`) relies on anyone calling `updateState()` to check conditions and advance phases. We considered integrating Chainlink Automation to guarantee timely transitions but decided against it.

Cost vs. Benefit. Chainlink Automation charges per execution (currently \$0.10–\$1.00 depending on gas prices and network) [6]. For a contract that might remain in `ACTIVE` state for years or decades, regular keeper pings would accumulate significant costs with no user benefit. Even monthly checks over 10 years would cost \$12–\$120, and the contract would need to be pre-funded to cover these fees.

Incentive Alignment. Beneficiaries have a direct financial incentive to call `updateState()` after the owner’s death: it is the only way to trigger distribution and receive their inheritance. Similarly, the notary (often a trusted family member or legal professional) can be instructed to monitor the contract. Since `updateState()` is public and gas costs are negligible compared to inheritance values, relying on interested parties is economically rational and removes recurring operational expenses.

Liveness Assumption. This design assumes at least one beneficiary or the notary will act when needed. For high-value estates, this is a reasonable assumption. Users requiring absolute automation (e.g., no living beneficiaries, all recipients are minors) could separately fund a keeper service, but we judged this edge case insufficient to justify baking automation costs into every deployment.

3.3 Aave Integration for Yield Generation

Funds deposited into the inheritance contract are supplied to an Aave-compatible lending pool to earn yield while awaiting distribution [2]. This decision stems from several factors:

Capital Efficiency. Inheritance timelines are inherently long — assets might sit in the contract for years or decades. Leaving funds idle represents a significant opportunity cost. By supplying assets to Aave, the contract earns interest (typically 2–6% APY on stablecoins like USDC), growing the inheritance value over time. For a \$100,000 deposit earning 4% over 10 years, this adds \$48,024 to the final distribution.

Security and Maturity. Aave V3 is one of the most battle-tested DeFi protocols, with over \$10 billion in total value locked and extensive security audits [2]. Its lending pools are non-custodial (the contract retains withdrawal rights) and have a strong track record. Using a proven primitive reduces implementation risk compared to building a custom yield strategy.

Liquidity and Composability. Aave supplies remain liquid: the contract can withdraw funds at any time for owner withdrawals or final distribution. This flexibility is critical—vesting or locked staking would prevent the owner from accessing their own funds. Aave’s `aToken` model (interest-bearing tokens) integrates cleanly with ERC-20 workflows.

Stablecoin Focus. The current implementation uses USDC (or a mock equivalent) to avoid volatility risk. Yield on stablecoins is lower than on volatile assets but ensures predictable distributions. Extending to ETH or other tokens would require additional risk disclosures and potentially dynamic allocation strategies, which we deferred to future work.

3.4 Single Notary vs. Multi-Signature Verification

The original proposal mentioned 2-of-3 trusted contacts for death verification. The implemented design uses a single notary address. This simplification reduces contract complexity (no threshold signature logic) and reflects a pragmatic trust model: users select one highly trusted entity (e.g., a lawyer, family member, or professional executor service). The notary's role is narrowly scoped to uploading death verification; they cannot withdraw funds or change beneficiaries, limiting abuse potential. Future versions could introduce multi-sig verification via OpenZeppelin's `AccessControl` or a separate oracle aggregator for users requiring decentralized attestation.

3.5 Scope Reductions and Future Work

Several features from the initial proposal were deferred:

3.6 Summary of Trade-offs

The implemented protocol represents a pragmatic subset of the initial vision, optimized for:

- **Core value delivery:** Solving the primary problem (crypto inheritance loss) without feature bloat.
- **Gas efficiency:** Bounded operations, minimal storage, no recurring costs.
- **Security surface:** Fewer features mean fewer attack vectors and simpler auditing.
- **User autonomy:** No mandatory off-chain dependencies or subscription fees.

Future work can layer additional features (vesting, multi-asset, decentralized oracles) as optional modules or through a factory pattern that deploys specialized contract variants based on user needs.

4 Tool usage / tech stack

This project spans two domains: on-chain smart contract development and an off-chain web client for interaction.

4.1 Core Languages

Solidity (v0.8.28) is used for smart contract logic, leveraging built-in overflow checks and modern language features [16]. **TypeScript** is used across both the contract testing layer and the frontend for static typing and improved tooling [17].

4.2 Smart Contract Framework and Tooling

Hardhat serves as the primary development and testing framework [10]. It provides a local Ethereum network, deterministic deployments, stack traces, and plugin extensibility. Scripts like `scripts/deploy.js` and `scripts/auto-deploy.js` automate contract deployment, while `start.js` boots a local node and performs an initial deployment for the frontend to consume.

Hardhat Ignition is adopted for more declarative deployment pipelines [11]. Ignition modules (e.g., in `ignition/modules/`) describe deployment intent, helping reduce manual sequencing errors and making deployments reproducible.

OpenZeppelin Contracts supplies audited base contracts (`Ownable`, `ReentrancyGuard`, `IERC20`) to reduce implementation risk and accelerate development [15]. Using well-established libraries mitigates common vulnerabilities and increases readability for reviewers.

4.3 Testing Stack

Unit and integration tests use **Mocha** as the test runner and **Chai** with the `chai-as-promised` pattern for expressive assertions [12, 5]. **Hardhat Network Helpers** assist with time skips to keep tests isolated and deterministic. The testing approach validates critical flows: beneficiary management, state machine transitions, death oracle integration, and payout distribution.

4.4 Linting and Code Quality

ESLint with the `Next.js` configuration enforces consistent style and catches common mistakes in the `React/TypeScript` codebase [9]. Static analysis complements `TypeScript`'s type checking by addressing stylistic and best-practice concerns (unused variables, unsafe `React` patterns). For `Solidity`, reliance on established `OpenZeppelin` components and compiler warnings keeps the contract surface maintainable; future work could add a dedicated static analysis pass (e.g., `Slither`) to further enhance assurance.

4.5 Runtime and Scripting

Node.js underpins `Hardhat` scripts, local tooling, and the custom automation scripts (`start.js`, `cleanup.js`) [14]. These scripts orchestrate development workflow: spinning up a local chain, deploying contracts, and cleaning artifacts to reset state between test sessions.

Appendices

```
1 // SPDX-License-Identifier: MIT
2 pragma solidity ^0.8.28;
3
4
5 import "@openzeppelin/contracts/access/Ownable.sol";
6 import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
7 import "@openzeppelin/contracts/utils/ReentrancyGuard.sol";
8 import {IDeathOracle} from "../mocks/IDeathOracle.sol";
9 import {MockAavePool} from "../mocks/MockAavePool.sol";
10
11 contract InheritanceProtocol is Ownable, ReentrancyGuard {
12
13     IERC20 public immutable usdc;
14     IDeathOracle public immutable deathOracle;
15     address private notaryAddress;
16     MockAavePool public aavePool;
17
18     // address for donations (underdetermined payout)
19     address private ourAddress;
20
21     /**
```



```

22     * Stores address and payout percentage amount (0-100) of
23     a beneficiary.
24     */
25     struct Beneficiary {
26         address payoutAddress;
27         uint256 amount;
28     }
29
30     Beneficiary[10] private _beneficiaries;
31
32     State private _currentState;
33
34     uint256 private _lastCheckIn;
35     bool private _called = false;
36
37     uint256 private constant NOT_FOUND = type(uint256).max;
38     uint256 private constant MAX_BENEFICIARIES = 10;
39     uint256 private constant MAX_PERCENTAGE = 100;
40     uint256 private constant CHECK_IN_PERIOD = 90 * 1 days;
41     uint256 private constant GRACE_PERIOD = 30 * 1 days;
42
43     event BeneficiaryAdded(address indexed payoutAddress,
44         uint256 amount, uint256 index);
45     event BeneficiaryRemoved(address indexed payoutAddress,
46         uint256 index);
47     event Deposited(uint256 amount);
48     event Withdrawn(uint256 amount);
49     event CheckedIn(uint256 timestamp);
50     event StateChanged(uint256 timestamp, State from, State
51         to);
52     event PayoutMade(uint256 amount, address payoutAddress);
53     event TestEvent(string s);
54     event TestEventNum(uint s);
55
56     /**
57     * Initializes a new InheritanceProtocol.
58     * @param _usdcAddress address of the currency used
59     * (non-zero).
60     */
61     constructor(address _usdcAddress, address
62         _deathOracleAddress, address _notaryAddress, address
63         _aavePoolAddress) Ownable(msg.sender) {
64         require(_usdcAddress != address(0), "USDC address
65             zero");
66         require(_deathOracleAddress != address(0), "Death
67             Oracle address zero");
68         ourAddress =
69             0xf39Fd6e51aad88F6F4ce6aB8827279cFfFb92266;
70         usdc = IERC20(_usdcAddress);
71         deathOracle = IDeathOracle(_deathOracleAddress);
72         notaryAddress = _notaryAddress;
73         aavePool = MockAavePool(_aavePoolAddress);
74         _currentState = State.ACTIVE;
75         _lastCheckIn = block.timestamp;
76     }
77
78     /// ----- MODIFIERS -----

```

```

70      /**
71       * This modifier requires the function call to be made
72       * before distribution.
73       */
74      modifier onlyPreDistribution() {
75          require(_currentState < State.DISTRIBUTION, "Cannot
76              modify funds post-distribution");
77          -;
78      }
79
80      /**
81       * This modifier requires the function call to be made in
82       * the ACTIVE or WARNING phase
83       */
84      modifier onlyActiveWarning() {
85          require(_currentState < State.VERIFICATION, "Cannot
86              make administrative changes without Owner
87              check-In");
88          -;
89      }
90
91      /**
92       * This modifier requires the function call to be made in
93       * the DISTRIBUTION phase
94       */
95      modifier onlyDistribution() {
96          require(_currentState == State.DISTRIBUTION, "Can only
97              make payouts in distribution phase");
98          -;
99      }
100
101      /**
102       * This modifier requires the function call to be made by
103       * the notary
104       */
105      modifier onlyNotary() {
106          require(msg.sender == notaryAddress, "Only notary can
107              call this function");
108          -;
109      }
110
111      /// ----- STATE MACHINE & CHECK-INS -----
112
113      /**
114       * Defines the state of the contract.
115       * - Active: mutable state, owner check-ins required.
116       * - Warning: Missed check-in, notification sent at 90
117       *   days,
118       *   verification phase starts at 120 days.
119       * - Verification: submission of death certificate (30
120       *   days).
121       * - Distribution: distribute assets based on defined
122       *   conditions.
123       */
124      enum State { ACTIVE, WARNING, VERIFICATION, DISTRIBUTION }
125
126      /**
127       * Updates the State in the State-Machine

```

```

116     * Should always be possible and accessible by anyone
117     * @return currentState after execution
118     */
119     function updateState() public returns (State) {
120         uint256 elapsed = uint256(block.timestamp) -
            _lastCheckIn;
121         State oldState = _currentState;
122
123         // --- Phase transitions in logical order ---
124
125         // If in ACTIVE and check-in expired WARNING
126         if (_currentState == State.ACTIVE && elapsed >
            CHECK_IN_PERIOD) {
127             _currentState = State.WARNING;
128         }
129
130         // If in WARNING and grace period expired
131         VERIFICATION
132         if (_currentState == State.WARNING && elapsed >
            CHECK_IN_PERIOD + GRACE_PERIOD) {
133             _currentState = State.VERIFICATION;
134         }
135
136         // If in VERIFICATION and death confirmed
137         DISTRIBUTION
138         if (_currentState == State.VERIFICATION &&
            deathOracle.isDeceased(owner())) {
139             _currentState = State.DISTRIBUTION;
140
141             emit StateChanged(block.timestamp, oldState,
                _currentState);
142
143             // Trigger payout if we reached DISTRIBUTION
144             if (_currentState == State.DISTRIBUTION) {
145                 distributePayout();
146             }
147
148             return _currentState;
149         }
150
151         /**
152         * Changes the state of the contract to a given state.
153         * @param to the state to change to.
154         */
155         function changeState (State to) public {
156             require(to != _currentState, "Already in requested
                state");
157             emit StateChanged(block.timestamp, _currentState, to);
158             _currentState = to;
159         }
160
161         /**
162         * The owner checks in to verify that he's alive.
163         * Should be possible in active and warning state.
164         */
165         function checkIn() public onlyOwner {

```

```

165         require(_currentState == State.ACTIVE || _currentState
166                == State.WARNING, "Need to be in active or warning
167                state");
168         emit CheckedIn(block.timestamp);
169         _lastCheckIn = block.timestamp;
170     }
171
172     /// ----- BENEFICIARY HANDLING -----
173
174     /**
175      * Finds the index of a beneficiary in the beneficiaries
176      * list.
177      * @param _address the address whose index to find.
178      * @return the index if the address is in the list,
179      * 'NOT_FOUND' otherwise.
180      */
181     function findBeneficiaryIndex(address _address) public
182     view returns (uint256) {
183         if (_address == address(0)) {
184             return NOT_FOUND;
185         }
186         for (uint256 i = 0; i < MAX_BENEFICIARIES; i++) {
187             if (_beneficiaries[i].payoutAddress == _address) {
188                 return i;
189             }
190         }
191         return NOT_FOUND;
192     }
193
194     /**
195      * Removes a beneficiary with a given address.
196      * Only the owner can perform this action.
197      * @param _address the address to remove.
198      * Fails if the provided address is zero OR not in the
199      * list of beneficiaries.
200      * @return true if the deletion was successful, false
201      * otherwise.
202      */
203     function removeBeneficiary(address _address) public
204     onlyOwner onlyActiveWarning returns (bool) {
205         checkIn();
206         uint256 index = findBeneficiaryIndex(_address);
207         if (index == NOT_FOUND) {
208             return false;
209         }
210         delete _beneficiaries[index];
211         emit BeneficiaryRemoved(_address, index);
212         return true;
213     }
214
215     /**
216      * Adds a beneficiary to the list.
217      * Only the owner can perform this action.
218      * Requirements:
219      * - List not full
220      * - Payout after adding <= 100
221      * @param _address the address to add to the list.

```

```

214     * @param _amount the payout amount related to this
215     address.
216     * @return true if the addition was successful, false
217     otherwise.
218     */
219     function addBeneficiary(address _address, uint256 _amount)
220     public onlyOwner onlyActiveWarning returns (bool) {
221         checkIn();
222         require(_address != address(0), "Invalid address");
223         require(_amount > 0 && _amount <= MAX_PERCENTAGE,
224             "Invalid amount");
225
226         // Check for duplicate
227         if (findBeneficiaryIndex(_address) != NOT_FOUND) {
228             return false;
229         }
230
231         uint256 currentSum = getDeterminedPayoutPercentage();
232         if (currentSum + _amount > MAX_PERCENTAGE) {
233             // it should not be possible to payout more than
234             // 100%
235             return false;
236         }
237
238         // Find empty slot
239         uint256 emptyIndex = NOT_FOUND;
240         for (uint256 i = 0; i < MAX_BENEFICIARIES; i++) {
241             if (_beneficiaries[i].payoutAddress == address(0)) {
242                 emptyIndex = i;
243                 break;
244             }
245         }
246
247         if (emptyIndex == NOT_FOUND) {
248             return false; // Max beneficiaries reached
249         }
250
251         _beneficiaries[emptyIndex] = Beneficiary({
252             payoutAddress: _address, amount: _amount });
253         emit BeneficiaryAdded(_address, _amount, emptyIndex);
254         return true;
255     }
256
257     /// ----- BALANCE HANDLING -----
258
259     /**
260     * Deposits a given amount of USDC.
261     * @param _amount the amount to deposit.
262     */
263     function deposit(uint256 _amount) external onlyOwner
264     nonReentrant onlyPreDistribution {
265         checkIn();
266         require(_amount > 0, "Amount has to be greater than
267             zero.");
268
269         usdc.transferFrom(msg.sender, address(this), _amount);

```

```

263         usdc.approve(address(aavePool), _amount);
264
265         aavePool.supply(address(usdc), _amount, address(this));
266
267         emit Deposited(_amount);
268     }
269
270     /**
271      * Withdraws a given amount of USDC.
272      * @param _amount the amount to withdraw.
273      */
274     function withdraw(uint256 _amount) external onlyOwner
275         nonReentrant onlyPreDistribution {
276         checkIn();
277         require(_amount > 0, "Amount has to be greater than
278             zero.");
279         require(getBalance() >= _amount, "Insufficient
280             balance");
281
282         aavePool.withdraw(address(usdc), _amount,
283             address(this));
284
285         usdc.transfer(msg.sender, _amount);
286         emit Withdrawn(_amount);
287     }
288
289     /// ----- DEATH CERTIFICATION -----
290
291     /**
292      * Upload the death verification to the chain
293      * Only callable by the notary
294      */
295     function uploadDeathVerification(bool _deceased, bytes
296         calldata _proof) external onlyNotary{
297         deathOracle.setDeathStatus(owner(), _deceased, _proof);
298     }
299
300     /**
301      * Checks if the owner died by calling death certificate
302      * oracle.
303      * @return true if the owner died, else otherwise.
304      */
305     function checkIfOwnerDied() public view returns (bool) {
306         return deathOracle.isDeceased(owner());
307     }
308
309     /// ----- DISTRIBUTION METHODS -----
310
311     /**
312      * Distributes the payout based on definitions given by
313      * owner.
314      * Is only called in the updateState() Function, after
315      * death verification
316      */
317     function distributePayout() public {
318         require(!_called, "Payout can only be called once.");
319         _called = true;
320         bool donation = !isPayoutFullyDetermined();

```

```

313         uint256 count = getActiveCount();
314         Beneficiary[] memory activeBeneficiaries =
            getActiveBeneficiaries();
315         uint256 balanceRemainingInPool = getBalance();
316         uint256 originalBalance =
            aavePool.withdraw(address(usdc),
            balanceRemainingInPool, address(this));
317         for (uint256 i=0; i<count; i++) {
318             Beneficiary memory beneficiary =
                activeBeneficiaries[i];
319             uint256 amount = beneficiary.amount;
320             address payoutAddress = beneficiary.payoutAddress;
321
322             uint actualAmount = (originalBalance * amount) /
                MAX_PERCENTAGE;
323
324             usdc.transfer(payoutAddress, actualAmount);
325             emit PayoutMade(actualAmount, payoutAddress);
326         }
327         if (donation) {
328             // If the payout is not fully determined, the rest
of the balance will be sent to the developer
team.
329             // For now this is hardcoded as the first address
generated by hardhat when running a local node.
330             uint256 donatedAmount =
                aavePool.withdraw(address(usdc), getBalance(),
                address(this));
331             usdc.transfer(ourAddress, donatedAmount);
332             emit PayoutMade(donatedAmount, ourAddress);
333         }
334     }
335
336     /// ----- VIEW METHODS -----
337
338     /**
339      * Checks if the currently defined payout is fully
 determined, meaning
340      * 100% of the balance is being spent.
341      * @return true if the full balance will be spent, false
 otherwise.
342     */
343     function isPayoutFullyDetermined() public view returns
        (bool) {
344         uint256 sum = getDeterminedPayoutPercentage();
345         return sum == MAX_PERCENTAGE;
346     }
347
348     /**
349      * Calculates the percentage amount of currently
 determined payout.
350      * @return a number between 0 and 100, equivalent to the
 combined relative payout.
351     */
352     function getDeterminedPayoutPercentage() public view
        returns (uint256) {
353         uint256 sum;
354         for (uint256 i = 0; i < MAX_BENEFICIARIES; i++) {

```

```

355         if (_beneficiaries[i].payoutAddress != address(0))
356         {
357             sum += _beneficiaries[i].amount;
358         }
359     return sum;
360 }
361
362 /**
363  * Gets the current balance.
364  * @return the balance of the combined deposited funds.
365  */
366 function getBalance() public view returns (uint256) {
367     return aavePool.getBalance(address(this));
368 }
369
370 /**
371  * Getter for the beneficiaries list.
372  * @return the list of 10 beneficiaries (might contain
373  * empty slots).
374  */
375 function getBeneficiaries() public view returns
376 (Beneficiary[10] memory) {
377     return _beneficiaries;
378 }
379
380 /**
381  * Counts the number of active beneficiaries.
382  * @return the number of active beneficiaries.
383  */
384 function getActiveCount() public view returns (uint256) {
385     uint256 count;
386     for (uint256 i = 0; i < MAX_BENEFICIARIES; i++) {
387         if (_beneficiaries[i].payoutAddress != address(0))
388         {
389             count++;
390         }
391     }
392     return count;
393 }
394
395 /**
396  * Gets only the active beneficiaries.
397  * @return an array of beneficiaries.
398  */
399 function getActiveBeneficiaries() public view returns
400 (Beneficiary[] memory) {
401     uint256 activeCount = getActiveCount();
402     Beneficiary[] memory active = new
403     Beneficiary[](activeCount);
404     uint256 count = 0;
405     for (uint256 i = 0; i < MAX_BENEFICIARIES; i++) {
406         if (_beneficiaries[i].payoutAddress != address(0))
407         {
408             active[count] = _beneficiaries[i];
409             count++;
410         }
411     }
412 }

```



```

406         return active;
407     }
408
409     /**
410      * Gets the current state of the contract.
411      * @return the current state.
412      */
413     function getState() public view returns (State) {
414         return _currentState;
415     }
416
417     /**
418      * Gets the last check-in time.
419      * @return the last check-in time.
420      */
421     function getLastCheckIn() public view returns (uint256) {
422         return _lastCheckIn;
423     }
424
425 }

```

Listing 2: smart contract

A References

The entire project can be found at the [13] at https://github.com/vinentschall/decentralized_inheritance_protocols

References

- [1] *A Broad Overview of Reentrancy Attacks in Solidity Contracts*. Accessed 2025-11-16. 2025. URL: <https://www.quicknode.com/guides/ethereum-development/smart-contracts/a-broad-overview-of-reentrancy-attacks-in-solidity-contracts>.
- [2] *Aave V3: Pool Contract and Supplying Liquidity*. Accessed 2025-11-16. 2025. URL: <https://docs.aave.com/developers/core-contracts/pool>.
- [3] Bidget. *How Many Bitcoin Have Been Lost?* Accessed 2025-11-06. 2025. URL: <https://www.bitget.com/wiki/how-many-bitcoin-have-been-lost>.
- [4] V. Buterin and F. Vogelsteller. *ERC-20: Token Standard*. Accessed 2025-11-16. 2015. URL: <https://eips.ethereum.org/EIPS/eip-20>.
- [5] *Chai Assertion Library*. Accessed 2025-11-16. 2025. URL: <https://www.chaijs.com/>.
- [6] *Chainlink Automation Documentation*. Accessed 2025-11-16. 2025. URL: <https://docs.chain.link/chainlink-automation/introduction>.
- [7] *Dynamic Arrays and its Operations in Solidity*. Accessed 2025-11-16. 2025. URL: <https://www.geeksforgeeks.org/solidity/dynamic-arrays-and-its-operations-in-solidity/>.
- [8] *EIP-2612: Permit — 712-signed approvals*. Accessed 2025-11-16. 2020. URL: <https://eips.ethereum.org/EIPS/eip-2612>.
- [9] *ESLint Documentation*. Accessed 2025-11-16. 2025. URL: <https://eslint.org/docs/latest/>.
- [10] *Hardhat Documentation*. Accessed 2025-11-16. 2025. URL: <https://hardhat.org/docs>.

- [11] *Hardhat Ignition Deployment Framework*. Accessed 2025-11-16. 2025. URL: <https://hardhat.org/ignition>.
- [12] *Mocha Test Framework Documentation*. Accessed 2025-11-16. 2025. URL: <https://mochajs.org/>.
- [13] Vincent Schall Noah Klaholz and Max Mendes Carvalho. *Decentralized Inheritance Protocol*. Repository. https://github.com/vinentschall/decentralized_inheritance_protocol. 2025.
- [14] *Node.js Documentation v20+*. Accessed 2025-11-16. 2025. URL: <https://nodejs.org/en/docs>.
- [15] *OpenZeppelin Contracts Documentation*. Accessed 2025-11-16. 2025. URL: <https://docs.openzeppelin.com/contracts/5.x/>.
- [16] *Solidity Documentation v0.8.28*. Accessed 2025-11-16. 2025. URL: <https://docs.soliditylang.org/en/v0.8.28/>.
- [17] *TypeScript Handbook*. Accessed 2025-11-16. 2025. URL: <https://www.typescriptlang.org/docs/handbook/intro.html>.